
Abstract. 1

Generic Haskell is an extension of Haskell that supports the construc-
tion of generic programs. This article describes generic programming in
practice. It discusses three advanced generic programming applications:
generic dictionaries, compressing XML documents, and the zipper. When
describing and implementing these examples, we will encounter some
advanced features of Generic Haskell, such as type-indexed data types,
dependencies between and generic abstractions of generic functions, ad-
justing a generic function using a default case, and generic functions with
a special case for a particular constructor.

1 Introduction

A polytypic (or generic, type-indexed) function is a function that can be instan-
tiated on many data types to obtain data type specific functionality. Examples
of polytypic functions are the functions that can be derived in Haskell [50], such
as show , read , and ‘ ’. In [23] we have introduced type-indexed functions, and
we have shown how to implement them in Generic Haskell [7]. For an older
introduction to generic programming, see Backhouse et al [4].

Why is generic programming important? Generic programming makes pro-
grams easier to write:

– Programs that could only be written in an untyped style can now be written
in a language with types.

– Some programs come for free.
– Some programs are simple adjustments of library functions, instead of com-

plicated traversal functions.

1 This is a preliminary version of the notes that will appear in the Lecture Notes of
the Summer School on Generic Programming.

Of course not all programs become simpler when you write your programs in
a generic programming language, but, on the other hand, no programs become
more complicated. In this paper we will try to give you a feeling about where
and when generic programs are useful.

This article describes three advanced generic programming applications: ge-
neric dictionaries, compressing XML documents (and XML tools in general),
and the zipper. The example applications are described in more detail below. In
the examples, we will encounter several new generic programming concepts:

– Type-indexed data types. A type-indexed data type is constructed in a generic
way from an argument data type [24]. It is the equivalent of type-indexed
functions on the level of data types.

– Default cases. To define a generic function that is the same as another func-
tion except for a few cases we use a default case [8]. If the new definition
does not provide a certain case, then the default case applies and copies the
case from another function.

– Constructor cases. A constructor case of a generic program deals with a
constructor of a data type that requires special treatment [8]. Constructor
cases are especially useful when dealing with data types with a large num-
ber of constructors, and only a small number of constructors need special
treatment.

– Dependencies and generic abstractions. To write a generic function that uses
another generic function we can use a dependency or a generic abstraction [8].

We will introduce these concepts where and when we need them.

Example 1: Digital searching. A digital search tree or trie is a search tree scheme
that employs the structure of search keys to organize information. Searching is
useful for various data types, so we would like to allow for keys and information
of any data type. This means that we have to construct a new kind of trie for
each key type. For example, consider the data type String defined by

data String = NilS | ConsS Char String ,

We can represent string-indexed tries with associated values of type v as follows:

data FMapString v = NullString
| TrieString (Maybe v) (FMapChar (FMapString v))

Such a trie for strings would typically be used for a concordance or another
index on texts. The first component of the constructor TrieString contains
the value associated with NilS . The second component of TrieString is de-
rived from the constructor ConsS :: Char → String → String . We assume
that a suitable data structure FMapChar and an associated look-up function
lookupChar :: ∀v .Char → FMapChar v → Maybe v for characters are prede-
fined. Given these prerequisites we can define a look-up function for strings as

follows:
lookupString :: String → FMapString v → Maybe v
lookupString NilS (TrieString tn tc) = tn
lookupString (ConsS c s) (TrieString tn tc) =

(lookupChar c 3 lookupString s) tc.

To look up a non-empty string, ConsS c s, we look up c in the FMapChar
obtaining a trie, which is then recursively searched for s. Since the look-up
functions have result type Maybe v , we use the monadic composition of the
Maybe monad, called ‘3’, to compose lookupString and lookupChar .

(3) :: (a → Maybe b)→ (b → Maybe c)→ a → Maybe c
(f 3 g) a = case f a of {Nothing → Nothing ; Just b → g b}.

In the following section we will show how to define a trie and an associated
look-up function for an arbitrary data type.

Example 2: Compressing XML documents. XML documents may become (very)
large because of the markup that is added to the content. Because of the repeti-
tive structure of many XML documents, these documents can be compressed by
quite a large factor.

An XML document is usually structured according to a DTD (Document
Type Definition), a specification that describes which tags may be used in the
XML document, and in which positions and order they have to be. A DTD is,
in a way, the type of an XML document. An XML document is called valid with
respect to a certain DTD if it follows the structure that is specified by that
DTD. An XML compressor can use information from the DTD to obtain better
compression. For example, consider the following small XML file:

<book lang="English">
<title> Dead Famous </title>
<author> Ben Elton </author>
<date> 2001 </date>
</book>

This file may be compressed by separating the structure from the data, and
compressing the two parts separately. For compressing the structure we can
make good use of the DTD. If we know how many elements, say n, appear in the
DTD (the DTD for the above document contains at least 4 elements), we can
replace each occurrence of the markup of an element in an XML file which is
valid with respect to the DTD by log2 n bits. This simple idea is the main idea
behind the tool described in Section 3, and has been described in the context of
data conversion by Jansson and Jeuring [31, 35].

In Section 3 we use HaXml [58] to translate a DTD to a data type, and
we construct generic functions for separating the contents (the strings) and the
shape (the constructors) of a value of a data type, and for encoding the shape
of a value of a data type using information about the (number of) constructors
of the data type.

XML compressors ar just one class of XML tools that are easily implemented
as generic programs. Other XML tools that can be implemented as generic pro-
grams are XML editors, XML databases, and XML version management tools.

Example 3: Zipper. The zipper [27] is a data structure that is used to represent
a tree together with a subtree that is the focus of attention, where that focus
may move left, right, up, or down the tree. For example, the data type Tree and
its corresponding zipper, called Loc Tree, are defined by

data Tree = Leaf Char | Fork Tree Tree
type Loc Tree = (Tree,Context Tree)
data Context Tree = top

| forkL Context Tree Tree
| forkR Tree Context Tree.

Using the type of locations Loc Tree we can efficiently navigate through a tree.
For example:

down Tree :: Loc Tree → Loc Tree
down Tree (Leaf a, c) = (Leaf a, c)
down Tree (Fork tl tr , c) = (tl , forkL c tr)
right Tree :: Loc Tree → Loc Tree
right Tree (tl , forkL c tr) = (tr , forkR tl c)
right Tree l = l .

The navigator function down Tree moves the focus of attention to the leftmost
subtree of the current node; right Tree moves the focus to its right sibling.

Huet [27] defines the zipper data structure for rose trees and for the data
type Tree, and gives the generic construction in words. In Section 4 we describe
the zipper in more detail and show how to define a zipper for an arbitrary data
type.

Other applications of generic programming. Besides the applications mentioned
in the examples above, there are several application areas in which generic pro-
gramming can be used.

– Haskell’s deriving construct. Haskell’s deriving construct is used to gener-
ate code for for example the equality function, and for functions for reading
and showing values of data types. Only the classes Eq, Ord, Enum, Bounded,
Show and Read can be derived. The definitions of most of the derived func-
tions can be found in the library of Generic Haskell.

– Compiler functions. Several functions that are used in compilers are generic
functions: garbage collectors, tracers, debuggers, etc.

– Typed term processing. Functions like pattern matching, term rewriting and
unification are generic functions, and have been implemented as generic func-
tions in [36, 33, 34].

The form and functionality of these applications is exactly determined by the
structure of the input data.

Maybe the most common applications of generic programming can be found
in functions that traverse data built from rich mutually-recursive data types with
many constructors, and which perform computations on a single (or a couple of)
constructor(s). For example, consider a function which traverses an abstract
syntax tree and returns the free variables in the tree. Only for the variable
constructor something special has to be done, in all other cases the variables
collected at the children have to be passed on to the parent. This function can be
defined as an instance of a Generic Haskell library function crush [45], together
with a special constructor case for variables [8].

Organization. The rest of this paper is organized as follows. Section 2 introduces
generic dictionaries, and implements them in Generic Haskell. Section 3 describes
how generic programming can be used to construct XML tools. In particular,
it describes XComprez, a compressor for XML documents. Section 4 develops
a generic zipper data structure. Finally, Section 5 summarizes the main points
and concludes.

2 Generic dictionaries

A trie is a search tree scheme that employs the structure of search keys to
organize information. Tries were originally devised as a means to represent a
collection of records indexed by strings over a fixed alphabet. Based on work by
Wadsworth and others, Connelly et al. [10] generalized the concept to permit
indexing by elements built according to an arbitrary signature. In this section
we go one step further and define tries and operations on tries generically for
arbitrary data types of arbitrary kinds, including parameterized and nested data
types.

The Generic Haskell code for this section can be downloaded from the appli-
cations page on http://www.generic-haskell.org/.

2.1 Introduction

The concept of a trie was introduced by Thue in 1912 as a means to represent
a set of strings, see [39]. In its simplest form a trie is a multiway branching tree

where each edge is labelled with a character. For example,
the set of strings {ear , earl , east , easy , eye } is represented by
the trie depicted on the right. Searching in a trie starts at the
root and proceeds by traversing the edge that matches the first
character, then traversing the edge that matches the second
character, and so forth. The search key is a member of the
represented set if the search stops in a node that is marked—
marked nodes are drawn as filled circles on the right. Tries
can also be used to represent finite maps. In this case marked

nodes additionally contain values associated with the strings.
Interestingly, the move from sets to finite maps is not a mere
variation of the scheme. As we shall see it is essential for the
further development.

On a more abstract level a trie itself can be seen as a composition of finite
maps. Each collection of edges descending from the same node constitutes a
finite map sending a character to a trie. With this interpretation in mind it is
relatively straightforward to devise an implementation of string-indexed tries. If
strings are defined by the following data type:

data String = NilS | ConsS Char String ,

we can represent string-indexed tries with associated values of type v as follows.

data FMapString v = NullString
| TrieString (Maybe v) (FMapChar (FMapString v))

Here, NullString represents the empty trie. The first component of the con-
structor TrieString contains the value associated with NilS . Its type is Maybe v
instead of v since NilS may not be in the domain of the finite map represented by
the trie. In this case the first component equals Nothing . The second component
corresponds to the edge map. To keep the introductory example manageable we
implement FMapChar using ordered association lists.

type FMapChar v = [(Char , v)]
lookupChar :: ∀v .Char → FMapChar v → Maybe v
lookupChar c [ ] = Nothing
lookupChar c ((c′, v) : x )
| c < c′ = Nothing
| c c′ = Just v
| c > c′ = lookupChar c x

Note that lookupChar has result type Maybe v . If the key is not in the domain
of the finite map, Nothing is returned.

Building upon lookupChar we can define a look-up function for strings. To
look up the empty string we access the first component of the trie. To look up
a non-empty string, say, ConsS c s we look up c in the edge map obtaining a
trie, which is then recursively searched for s.

lookupString :: ∀v .String → FMapString v → Maybe v
lookupString s NullString = Nothing
lookupString NilS (TrieString tn tc) = tn
lookupString (ConsS c s) (TrieString tn tc) =

(lookupChar c 3 lookupString s) tc

In the last equation we use monadic composition to take care of the error signal
Nothing .

Based on work by Wadsworth and others, Connelly et al. [10] have gener-
alized the concept of a trie to permit indexing by elements built according to
an arbitrary signature, that is, by elements of an arbitrary non-parameterized
data type. The definition of lookupString already gives a clue what a suitable
generalization might look like: the trie TrieString tn tc contains a finite map
for each constructor of the data type String ; to look up ConsS c s the look-up
functions for the components, c and s, are composed. Generally, if we have a
data type with k constructors, the corresponding trie has k components. To look
up a constructor with n fields, we must select the corresponding finite map and
compose n look-up functions of the appropriate types. If a constructor has no
fields (such as NilS ), we extract the associated value.

As a second example, consider the data type of external search trees:

data Dict = Leaf String | Node Dict String Dict .

A trie for external search trees represents a finite map from Dict to some value
type v . It is an element of FMapDict v given by

data FMapDict v = NullDict
| TrieDict (FMapString v)

(FMapDict (FMapString (FMapDict v))).

Note that FMapDict is a nested data type, since the recursive call on the right
hand side, FMapDict (FMapString (FMapDict v)), is a substitution instance of
the left hand side. Consequently, the look-up function on external search trees
requires polymorphic recursion.

lookupDict :: ∀v .Dict → FMapDict v → Maybe v
lookupDict d NullDict = Nothing
lookupDict (Leaf s) (TrieDict tl tn) = lookupString s tl
lookupDict (Node l s r) (TrieDict tl tn) =

(lookupDict l 3 lookupString s 3 lookupDict r) tn

Looking up a node involves two recursive calls. The first, lookupDict l , is of
type Dict → FMapDict X → Maybe X where X = FMapString (FMapDict v),
which is a substitution instance of the declared type.

Note that it is absolutely necessary that FMapDict and lookupDict are para-
metric with respect to the codomain of the finite maps. Had we restricted the
type of lookupDict to Dict → FMapDict T → T for some fixed type T , the
definition would have no longer type-checked. This also explains why the con-
struction does not work for the finite set abstraction.

Generalized tries make a particularly interesting application of generic pro-
gramming. The central insight is that a trie can be considered as a type-indexed
data type. This makes it possible to define tries and operations on tries gener-
ically for arbitrary data types. We already have the necessary prerequisites at
hand: we know how to define tries for sums and for products. A trie for a sum
is essentially a product of tries and a trie for a product is a composition of tries.

The extension to arbitrary data types is then uniquely defined. Mathematically
speaking, generalized tries are based on the following isomorphisms.

1→fin v ∼= v
(t1 + t2)→fin v ∼= (t1 →fin v) × (t2 →fin v)
(t1 × t2)→fin v ∼= t1 →fin (t2 →fin v)

Here, t →fin v denotes the set of all finite maps from t to v . Note that t →fin v
is sometimes written v [t ], which explains why these equations are also known as
the ‘laws of exponentials’.

2.2 Signature

To put the above idea in concrete terms we will define a type-indexed data type
FMap, which has the following type for types t of kind ?.

FMap〈t :: ?〉 :: ?→ ?,

So FMap assigns a type constructor of kind ?→ ? to each key type t of kind ?.
We will implement the following operations on tries.

empty〈t〉 :: ∀v .FMap〈t〉 v
isempty〈t〉 :: ∀v .FMap〈t〉 v → Bool
single〈t〉 :: ∀v . t × v → FMap〈t〉 v
lookup〈t〉 :: ∀v . t → FMap〈t〉 v → Maybe v
insert〈t〉 :: ∀v . (v → v → v)→ t × v → (FMap〈t〉 v → FMap〈t〉 v)
merge〈t〉 :: ∀v . (v → v → v)→ (FMap〈t〉 v → FMap〈t〉 v → FMap〈t〉 v)
delete〈t〉 :: ∀v . t → (FMap〈t〉 v → FMap〈t〉 v)

The value empty〈t〉 is the empty trie. Function isempty〈t〉 takes a trie and de-
termines whether or not it is empty. Function single〈t〉 (t , v) constructs a trie
that contains the binding (t , v) as its only element. The function lookup〈t〉 takes
a key and a trie and looks up the value associated with the key. The function
insert〈t〉 inserts a new binding into a trie and merge〈t〉 combines two tries.
Function delete〈t〉 takes a key and a trie, and removes the binding for the key
from the trie. The two functions insert〈t〉 and merge〈t〉 take as a first argument
a so-called combining function, which is applied whenever two bindings have the
same key. For instance, λnew old → new is used as the combining function for
insert〈t〉 if the new binding is to override an old binding with the same key. For
finite maps of type FMap〈t〉 Int addition may also be a sensible choice. Inter-
estingly, we will see that the combining function is not only a convenient feature
for the user; it is also necessary for defining insert〈t〉 and merge〈t〉 generically
for all types!

2.3 Type-indexed tries

We have already noted that generalized tries are based on the laws of exponen-
tials.

1→fin v ∼= v
(t1 + t2)→fin v ∼= (t1 →fin v) × (t2 →fin v)
(t1 × t2)→fin v ∼= t1 →fin (t2 →fin v)

In order to define the notion of finite map it is customary to assume that each
value type v contains a distinguished element or base point ⊥v , see [10]. A finite
map is then a function whose value is ⊥v for all but finitely many arguments. For
the implementation of tries it is, however, inconvenient to make such a strong
assumption (though one could use type classes for this purpose). Instead, we
explicitly add a base point when necessary motivating the following definition
of FMap:

type FMap〈Unit〉 v = FMUnit (Maybe v)
type FMap〈Char〉 v = FMChar (FMapChar v)
type FMap〈Int〉 v = FMInt (Patricia.Dict v)
type FMap〈:+:〉 fma fmb v = FMEither (fma v ×• fmb v)
type FMap〈:*:〉 fma fmb v = FMProd (fma (fmb v))
type FMap〈Con〉 fma v = FMCon (fma v)
type FMap〈Label〉 fma v = FMLabel (fma v)

Here, (×•) is the type of optional pairs.

data a ×• b = Null | Pair a b

Instead of optional pairs we can also use ordinary pairs in the definition of FMap:

type FMap〈:+:〉 fma fmb v = FMEither (fma v × fmb v).

This representation has, however, two major drawbacks: (i) it relies in an essen-
tial way on lazy evaluation and (ii) it is inefficient, see [21].

We assume there exists a suitable library implementing finite maps with
integer keys. Such a library could be based, for instance, on a data structure
known as a Patricia tree [49]. This data structure fits particularly well in the
current setting since Patricia trees are a variety of tries. For clarity, we will use
qualified names when referring to entities defined in the hypothetical module
Patricia.

FMap is a type-indexed data type. We introduce type-indexed data types by
example, for more background and theory, see [24]. Note that in each line of the
definition of FMap we define a constructor name such as for example FMUnit
in the Unit case, which can be used to construct elements of the type-indexed
data type.

Furthermore, in contrast with type-indexed functions, the constructor index
Con doesn’t mention a constructor description anymore. This is because a type
cannot depend on the value, so the constructor description can never be used in
the definition of a type-indexed data type.

Type-indexed data types should also be defined on Label , which is used to
represent a record in a data type. Since the definition of a type-indexed data type
on Label is almost always exactly the same as the definition of the type-indexed
data type on Con, we will almost always omit the definition of type-indexed
data types (and functions) on Label .

Since the trie for the unit type is given by Maybe v rather than v itself,
tries for isomorphic types are, in general, not isomorphic. We have, for instance,
Unit ∼= Unit :*:Unit (ignoring laziness) but FMap〈Unit〉 v = Maybe v 6∼=
Maybe (Maybe v) = FMap〈Unit :*:Unit〉 v . The trie type Maybe (Maybe v) has
two different representations of the empty trie: Nothing and Just Nothing . How-
ever, only the first one will be used in our implementation. Similarly, Maybe v ×•
Maybe v has two elements, Null and Pair Nothing Nothing , that represent the
empty trie. Again, only the first one will be used.

As mentioned in Section 2.2, the type of FMap for types of kind ? is ?→ ?.
For type constructors with higher-order kinds, the type of FMap looks surpris-
ingly similar to the type of type-indexed functions for higher-order kinds. A trie
on the type List a is a trie for the type List , applied to a trie for the type a:

FMap〈f :: ?→ ?〉 :: (?→ ?)→ (?→ ?)

The ‘type’ of a type-indexed type is a kind-indexed kind. In general, we have:

FMap〈f :: κ〉 :: FMAP〈κ〉 f
FMAP〈κ :: 2〉 :: 2

FMAP〈?〉 = ?→ ?
FMAP〈κ→ ν〉 = FMAP〈κ〉 → FMAP〈ν〉

Example 1. Let us specialize FMap to the following data types.

data List a = Nil | Cons a (List a)
data Tree a b = Leaf a | Node (Tree a b) b (Tree a b)
data Fork a = ForkF a a
data Sequ a = EndS | ZeroS (Sequ (Fork a)) | OneS a (Sequ (Fork a))

Recall that these types are represented by

List = Fix (ΛList . Λa .Unit :+: a :*:List a)
Tree = Fix (ΛTree . Λa b . a :+:Tree a b :*: b :*:Tree a b)
Fork = Λa . a :*: a
Sequ = Fix (ΛSequ . Λa .Unit :+:Sequ (Fork a) :+: a :*:Sequ (Fork a)).

Note that :*: binds stronger than :+:. Consequently, the corresponding trie
types are

FMapList = Fix (ΛFMapList . Λfa .Maybe ×• fa · FMapList fa)
FMapTree = Fix (ΛFMapTree . Λfa fb .

fa ×•
FMapTree fa fb · fb · FMapTree fa fb)

FMapFork = Λfa . fa · fa
FMapSequ = Fix (ΛFMapSequ . Λfa .

Maybe ×•
FMapSequ (FMapFork fa) ×•
fa · FMapSequ (FMapFork fa)).

As an aside, note that we interpret a1 ×• a2 ×• a3 as the type of optional triples
and not as nested optional pairs:

data a1 ×• a2 ×• a3 = Null | Triple a1 a2 a3.

Now, since Haskell permits the definition of higher-order kinded data types,
the second-order type constructors above can be directly coded as data types.
All we have to do is to bring the equations into an applicative form.

data FMapList fa v = NullList
| TrieList (Maybe v)

(fa (FMapList fa v))
data FMapTree fa fb v = NullTree

| TrieTree (fa v)
(FMapTree fa fb

(fb (FMapTree fa fb v)))

These types are the parametric variants of FMapString and FMapDict defined
in Section 2.1: we have FMapString ≈ FMapList FMapChar (corresponding
to String ≈ List Char) and FMapDict ≈ FMapTree FMapString FMapString
(corresponding to Dict ≈ Tree String String). Things become interesting if we
consider nested data types.

data FMapFork fa v = TrieFork (fa (fa v))
data FMapSequ fa v = NullSequ

| TrieSequ (Maybe v)
(FMapSequ (FMapFork fa) v)
(fa (FMapSequ (FMapFork fa) v))

The generalized trie of a nested data type is a second-order nested data type! A
nest is termed second-order, if a parameter that is instantiated in a recursive call
ranges over type constructors of first-order kind. The trie FMapSequ is a second-
order nest since the parameter fa of kind ?→ ? is changed in the recursive calls.
By contrast, FMapTree is a first-order nest since its instantiated parameter v

has kind ?. It is quite easy to produce generalized tries that are both first- and
second-order nests. If we swap the components of Sequ’s third constructor—
OneS a (Sequ (Fork a)) becomes OneS (Sequ (Fork a)) a—then the third
component of FMapSequ has type FMapSequ (FMapFork fa) (fa v) and since
both fa and v are instantiated, FMapSequ is consequently both a first- and a
second-order nest.

2.4 Empty tries

The empty trie is defined as follows.

type Empty〈〈?〉〉 t = ∀v .FMap〈t〉 v
type Empty〈〈κ→ ν〉〉 t = ∀a .Empty〈〈κ〉〉 a → Empty〈〈ν〉〉 (t a)
empty〈t :: κ〉 :: Empty〈〈κ〉〉 t
empty〈Unit〉 = FMUnit Nothing
empty〈Char〉 = FMChar [ ]
empty〈Int〉 = FMInt Patricia.empty
empty〈:+:〉 ea eb = FMEither Null
empty〈:*:〉 ea eb = FMProd ea
empty〈Con c〉 ea = FMCon ea

The definition already illustrates several interesting aspects of programming with
generalized tries. First, the explicit polymorphic type of empty is necessary to
make the definition work. Consider the line empty〈:*:〉 ea eb, which is of type
∀v .FMap〈t1〉 (FMap〈t2〉 v) for some t1 and t2. It is defined in terms of ea, which
is of type ∀v .FMap〈t1〉 v . That means that ea is used polymorphically. In other
words, empty makes use of polymorphic recursion!

Example 2. Let us specialize empty to lists and binary random-access lists.

emptyList :: ∀fa . (∀w . fa w)→ (∀v .FMapList fa v)
emptyList ea = NullList
emptyFork :: ∀fa . (∀w . fa w)→ (∀v .FMapFork fa v)
emptyFork ea = TrieFork ea
emptySequ :: ∀fa . (∀w . fa w)→ (∀v .FMapSequ fa v)
emptySequ ea = NullSequ

The second function, emptyFork , illustrates the polymorphic use of the param-
eter: ea has type ∀w . fa w but is used as an element of fa (fa w). The functions
emptyList and emptySequ show that the ‘mechanically’ generated definitions can
sometimes be slightly improved: the argument ea is not needed.

Function isempty〈t〉 takes a trie and determines whether or not it is empty.

type IsEmpty〈〈?〉〉 t = ∀v .FMap〈t〉 v → Bool
type IsEmpty〈〈κ→ ν〉〉 t = ∀a . IsEmpty〈〈κ〉〉 a → IsEmpty〈〈ν〉〉 (t a)

isempty〈t :: κ〉 :: IsEmpty〈〈κ〉〉 t
isempty〈Unit〉 (FMUnit v) = isNothing v
isempty〈Char〉 (FMChar l) = null l
isempty〈Int〉 (FMInt l) = Patricia.isempty l
isempty〈:+:〉 iea ieb (FMEither Null) = True
isempty〈:+:〉 iea ieb (FMEither d) = False
isempty〈:*:〉 iea ieb (FMProd d) = iea d
isempty〈Con c〉 iea (FMCon d) = iea d

Example 3. Let us specialize isempty to lists and binary random-access lists.

isemptyList :: ∀fa . (∀w . fa w → Bool)→
(∀v .FMapList fa v → Bool)

isemptyList iea NullList = True
isemptyList iea (TrieList tn tc) = False
isemptyFork :: ∀fa . (∀w . fa w → Bool)→

(∀v .FMapFork fa v → Bool)
isemptyFork iea (TrieFork tf ) = iea tf
isemptySequ :: ∀fa . (∀w . fa w → Bool)→

(∀v .FMapSequ fa v → Bool)
isemptySequ iea NullSequ = True
isemptySequ iea (TrieSequ tv tf ts) = False

2.5 Singleton tries

Function single〈t〉 (t , v) constructs a trie that contains the binding (t , v) as its
only element. To construct a trie in the sum case, we have to return a Pair , of
which only one component is inhabited. The other component is the empty trie.
This implies that single depends on empty , which in Generic Haskell is denoted
by

dependency single ← single empty

This line says that the generic function single depends on both empty and itself.
The right-hand side of the arrow ← enumerates the functions on which the left-
hand side argument depends. The effect of this dependency is that we can use
both the empty trie and the single trie of the children in the sum, product, and
constructor cases of function single. On higher-order kinds, the dependency on
function empty is reflected in the type of function single.

type Single〈〈?〉〉 t = ∀v . (t , v)→ FMap〈t〉 v
type Single〈〈κ→ ν〉〉 t = ∀a .Single〈〈κ〉〉 a → Empty〈〈κ〉〉 a → Single〈〈ν〉〉 (t a)

Plain generic functions can be seen as catamorphisms [42, 46] over the structure
of data types. With dependencies, we also get the power of paramorphisms [44]

(or, more general, even zygomorphisms [41]).

single〈t :: κ〉 :: Single〈〈κ〉〉 t
single〈Unit〉 (Unit , v) = FMUnit (Just v)
single〈Char〉 (c, v) = FMChar [(c, v)]
single〈Int〉 (i , v) = FMInt (Patricia.single (i , v))
single〈:+:〉 sa ea sb eb (Inl a, v) = FMEither (Pair (sa (a, v)) eb)
single〈:+:〉 sa ea sb eb (Inr b, v) = FMEither (Pair ea (sb (b, v)))
single〈:*:〉 sa ea sb eb (a :*: b, v) = FMProd (sa (a, sb (b, v)))
single〈Con d〉 sa ea (Con b, v) = FMCon (sa (b, v))

Example 4. Let us again specialize the generic function to lists and binary
random-access lists.

singleList :: ∀kT fa . (∀w . fa w)→ (∀w . kT × w → fa w)
→ (∀v . (List kT × v → FMapList fa v))

singleList ea sa (Nil , v) = TrieList (Just v) ea
singleList ea sa (Cons k ks, v) = TrieList Nothing (sa (k , singleList ea sa (ks, v)))
singleFork :: ∀kT fa . (∀w . fa w)→ (∀w . kT × w → fa w)

→ (∀v . (Fork kT × v → FMapFork fa v))
singleFork ea sa (ForkF k1 k2, v) = TrieFork (sa (k1, sa (k2, v)))
singleSequ :: ∀kT fa . (∀w . fa w)→ (∀w . kT × w → fa w)

→ (∀v . (Sequ kT × v → FMapSequ fa v))
singleSequ ea sa (EndS , v) = TrieSequ (Just v) NullSequ ea
singleSequ ea sa (ZeroS s, v)

= TrieSequ Nothing (singleSequ (emptyFork ea) (singleFork ea sa) (s, v)) ea
singleSequ ea sa (OneS k s, v)

= TrieSequ Nothing NullSequ (sa (k , singleSequ (emptyFork ea) (singleFork ea sa) (s, v)))

Again, we can simplify the ‘mechanically’ generated definitions: since the defini-
tion of Fork does not involve sums, singleFork does not require its first argument,
ea, which can be safely removed.

2.6 Look up

The look-up function implements the scheme discussed in Section 2.1.

type Lookup〈〈?〉〉 t = ∀v . t → FMap〈t〉 v → Maybe v
type Lookup〈〈κ→ ν〉〉 t = ∀a .Lookup〈〈κ〉〉 a → Lookup〈〈ν〉〉 (t a)

lookup〈t :: κ〉 :: Lookup〈〈κ〉〉 t
lookup〈Unit〉 Unit (FMUnit fm) = fm
lookup〈Char〉 c (FMChar fm) = lookupChar c fm
lookup〈Int〉 i (FMInt fm) = Patricia.lookup i fm
lookup〈:+:〉 la lb (Inl a) (FMEither (Pair fma fmb)) = la fma a
lookup〈:+:〉 la lb (Inr b) (FMEither (Pair fma fmb)) = lb fmb b
lookup〈:*:〉 la lb (a :*: b) (FMProd fma) = do fmb ← la fma a

lb fmb b
lookup〈Con d〉 l (Con b) (FMCon fm) = l fm b

On sums the look-up function selects the appropriate map; on products it ‘com-
poses’ the look-up functions for the components. Since lookup has result type
Maybe v , we use the monadic composition.

Suppose we want to define a function specialLookup that is almost the same
as function lookup, but uses a different function, say lookupCharEfficient , when
looking up characters. Then we can use a default case [8] to define function
specialLookup using function lookup as follows:

specialLookup〈t :: κ〉 :: Lookup〈〈κ〉〉 t
specialLookup〈Char〉 c (FMChar fm) = lookupCharEfficient c fm
specialLookup〈a〉 = lookup〈a〉

So function specialLookup is equal to the function lookup in all cases except for
the Char case, where it uses a special lookup function.

We also might want to use the special lookup function for just one kind of
characters that appear in a particular data type. For example, suppose we have
the following data type:

data C = C1 Char | C2 Char | ...

and we want to use the efficient lookup function lookupCharEfficient only for
characters under the constructor C1 . Then we can use a constructor case [8] to
define function specialLookup.

specialLookup〈t :: κ〉 :: Lookup〈〈κ〉〉 t
specialLookup〈case C1 〉 (C1 c) (FMChar fm) = lookupCharEfficient c fm
specialLookup〈a〉 = lookup〈a〉

Function specialLookup is still a generic function, but on values of the form C1 c
of type C it uses a different lookup function.

Example 5. Specializing lookup〈K 〉 to concrete instances of K is by now proba-
bly a matter of routine. We obtain

lookupList :: ∀kT fa . (∀w . kT → fa w → Maybe w)
→ (∀v .List kT → FMapList fa v → Maybe v)

lookupList la ks NullList = Nothing
lookupList la Nil (TrieList tn tc) = tn
lookupList la (Cons k ks) (TrieList tn tc) = (la k 3 lookupList la ks) tc
lookupFork :: ∀kT fa . (∀w . kT → fa w → Maybe w)

→ (∀v .Fork kT → FMapFork fa v → Maybe v)
lookupFork la (ForkF k1 k2) (TrieFork tf ) = (la k1 3 la k2) tf
lookupSequ :: ∀fa kT . (∀w . kT → fa w → Maybe w)

→ (∀v .Sequ kT → FMapSequ fa v → Maybe v)
lookupSequ la s NullSequ = Nothing
lookupSequ la EndS (TrieSequ te tz to) = te
lookupSequ la (ZeroS s) (TrieSequ te tz to) = lookupSequ (lookupFork la) s tz
lookupSequ la (OneS a s) (TrieSequ te tz to) = (la a 3 lookupSequ (lookupFork la) s) to

The function lookupList generalizes lookupString defined in Section 2.1; we have
lookupString ≈ lookupList lookupChar .

2.7 Inserting and merging

Insertion is defined in terms of merge and single.

insert〈t :: ?〉 :: (v → v → v)→ (t , v)→ FMap〈t〉 v → FMap〈t〉 v
insert〈t〉 c (x , v) d = merge〈t〉 c (single〈t〉 (x , v)) d

Function insert is defined as a generic abstraction. A generic abstraction lifts
the restrictions that are normally imposed on the type of a generic function.
For example, normally a type constructor of kind ? → ? is always translated
to a function by the translation scheme of generic functions. When you use a
generic abstraction, this can be circumvented. The abstracted type parameter
is, however, restricted to types of a fixed kind. In the above case, insert only
works for types of kind ?. In the exercise at the end of this section you will define
insert as a type-indexed function that works for type constructors of all kinds.

Merging two tries is surprisingly simple. Given an auxiliary function for com-
bining two values of type Maybe:

combine :: ∀v . (v → v → v)→
(Maybe v → Maybe v → Maybe v)

combine c Nothing Nothing = Nothing
combine c Nothing (Just v2) = Just v2

combine c (Just v1) Nothing = Just v1

combine c (Just v1) (Just v2) = Just (c v1 v2)

and a function for merging two association lists

mergeChar :: ∀v . (v → v → v)→
(FMapChar v → FMapChar v → FMapChar v)

mergeChar c [ ] x ′ = x ′

mergeChar c x [ ] = x
mergeChar c ((k , v) : x ) ((k ′, v ′) : x ′)

| k < k ′ = (k , v) : mergeChar c x ((k ′, v ′) : x ′)
| k k ′ = (k , c v v ′) : mergeChar c x x ′

| k > k ′ = (k ′, v ′) : mergeChar c ((k , v) : x ) x ′,

we can define merge as follows.

type Merge〈〈?〉〉 t = ∀v .
(v → v → v)→ FMap〈t〉 v → FMap〈t〉 v → FMap〈t〉 v

type Merge〈〈κ→ ν〉〉 t = ∀a .Merge〈〈κ〉〉 a → Merge〈〈ν〉〉 (t a)

merge〈t :: k〉 :: Merge〈〈k〉〉 t
merge〈Unit〉 c (FMUnit v) (FMUnit v ′) = FMUnit (combine c v v ′)
merge〈Char〉 c (FMChar fm) (FMChar fm ′) = FMChar (mergeChar fm ′ fm)
merge〈Int〉 c (FMInt fm) (FMInt fm ′) = FMInt (Patricia.mergeInt fm ′ fm)
merge〈Con d〉 ma c (FMCon e) (FMCon e ′) = FMCon (ma c e e ′)

For the sum case, we have to distinguish all possible forms of the tries to be
merged:

merge〈:+:〉 ma mb c d (FMEither Null) = d
merge〈:+:〉 ma mb c (FMEither Null) d = d
merge〈:+:〉 ma mb c (FMEither (Pair x y)) (FMEither (Pair v w)) =

FMEither (Pair (ma c x v) (mb c y w))

The most interesting equation is the product case. The tries d and d ′ are of
type FMap〈t1〉 (FMap〈t2〉 v), for some types t1 and t2. To merge them we can
recursively call ma; we must, however, supply a combining function of type
∀v .FMap〈t2〉 v → FMap〈t2〉 v → FMap〈t2〉 v . A moment’s reflection reveals
that mb c is the desired combining function.

merge〈:*:〉 ma mb c (FMProd d) (FMProd d ′) = FMProd (ma (mb c) d d ′)

The definition of merge shows that it is sometimes necessary to implement op-
erations more general than immediately needed. If Merge〈〈?〉〉 t had been the
simpler type ∀v .FMap〈t〉 v → FMap〈t〉 v → FMap〈t〉 v , then we would not
have been able to give a defining equation for :*:.

Example 6. To complete the picture let us again specialize the merging operation
for lists and binary random-access lists. The different instances of merge are
surprisingly concise (only the types look complicated).

mergeList :: ∀fa . (∀w . (w → w → w)→ (fa w → fa w → fa w))
→ (∀v . (v → v → v)
→ (FMapList fa v → FMapList fa v → FMapList fa v))

mergeList ma c NullList t = t
mergeList ma c t NullList = t
mergeList ma c (TrieList tn tc) (TrieList tn ′ tc′)

= TrieList (combine c tn tn ′)
(ma (mergeList ma c) tc tc′)

mergeFork :: ∀fa . (∀w . (w → w → w)→ (fa w → fa w → fa w))
→ (∀v . (v → v → v)
→ (FMapFork fa v → FMapFork fa v → FMapFork fa v))

mergeFork ma c (TrieFork tf ) (TrieFork tf ′)
= TrieFork (ma (ma c) tf tf ′)

mergeSequ :: ∀fa . (∀w . (w → w → w)→ (fa w → fa w → fa w))
→ (∀v . (v → v → v)
→ (FMapSequ fa v → FMapSequ fa v → FMapSequ fa v))

mergeSequ ma c NullSequ t = t
mergeSequ ma c t NullSequ = t
mergeSequ ma c (TrieSequ te tz to) (TrieSequ te ′ tz ′ to′)

= TrieSequ (combine c te te ′)
(mergeSequ (mergeFork ma) c tz tz ′)
(ma (mergeSequ (mergeFork ma) c) to to′) ut

2.8 Deleting

Function delete〈t〉 takes a key and a trie, and removes the binding for the key
from the trie. For the Char case we need a help function that removes an element
from an association list:

deleteChar :: ∀v .Char → FMapChar v → FMapChar v

and similarly for the Int case. Function delete is defined as follows:

delete〈t :: κ〉 :: Delete〈〈κ〉〉 t
delete〈Unit〉 Unit (FMUnit v) = FMUnit Nothing
delete〈Char〉 c (FMChar fm) = FMChar (deleteChar c fm)
delete〈Int〉 i (FMInt fm) = FMInt (Patricia.delete i fm)

All delete cases except the product case are relatively straightforward. In the
product case, we have to remove a binding for a product a :*: b. We do this by
using a to lookup the trie d in which there is a binding for b. Then we remove the
binding for b in d , obtaining a trie d ′. If d ′ is empty, then we delete the complete
binding for a in d , otherwise we insert the binding (a, d ′) in the original trie d .
Here we assume insertion overwrites existing bindings in a trie. Function delete
depends on functions lookup, insert , and isempty :

dependency delete ← delete lookup insert isempty

Here we need the kind-indexed typed version of function insert , as defined in
the exercise at the end of this section.

type Delete〈〈?〉〉 t = ∀v . t → FMap〈t〉 v → FMap〈t〉 v
type Delete〈〈κ→ ν〉〉 t = ∀a .Delete〈〈κ〉〉 a

→ Lookup〈〈κ〉〉 a
→ Insert〈〈κ〉〉 a
→ IsEmpty〈〈κ〉〉 a
→ Delete〈〈ν〉〉 (t a)

delete〈:+:〉 da la ia iea db lb ib ieb (Inl a) (FMEither (Pair x y)) =
FMEither (Pair (da a x ) y)

delete〈:+:〉 da la ia iea db lb ib ieb (Inr b) (FMEither (Pair x y)) =
FMEither (Pair x (db b y))

delete〈:*:〉 da la ia iea db lb ib ieb (a :*: b) (FMProd d) =
let Just d ′ = la a d

d ′′ = db b d ′

in if ieb d ′′ then FMProd (da a d) else FMProd (ia (a, d ′′) d)
delete〈Con c〉 da la ia iea (Con b) (FMCon d) =

FMCon (da b d)

2.9 Properties

The functions on tries enjoy several properties which hold generically for all
instances of t and which can be proved by fixed point induction.

lookup〈t〉 k (empty〈t〉) = Nothing
lookup〈t〉 k (single〈t〉 (k1, v1)) = if k k1 then Just v1 else Nothing
lookup〈t〉 k (merge〈t〉 c t1 t2) = combine c (lookup〈t〉 k t1) (lookup〈t〉 k t2)

The last law, for instance, states that looking up a key in the merge of two tries
yields the same result as looking up the key in each trie separately and then
combining the results. If the combining form c is associative,

c v1 (c v2 v3) = c (c v1 v2) v3,

then merge〈t〉 c is associative, as well. Furthermore, empty〈t〉 is the left and the
right unit of merge〈t〉 c:

merge〈t〉 c (empty〈t〉) x = x
merge〈t〉 c x (empty〈t〉) = x

merge〈t〉 c x1 (merge〈t〉 c x2 x3) = merge〈t〉 c (merge〈t〉 c x1 x2) x3.

2.10 Related work

Knuth [39] attributes the idea of a trie to Thue who introduced it in a pa-
per about strings that do not contain adjacent repeated substrings [54]. De la
Briandais [13] recommended tries for computer searching. The generalization of
tries from strings to elements built according to an arbitrary signature was dis-
covered by Wadsworth [57] and others independently since. Connelly et al. [10]
formalized the concept of a trie in a categorical setting: they showed that a trie
is a functor and that the corresponding look-up function is a natural transfor-
mation.

The first implementation of generalized tries was given by Okasaki in his
recent textbook on functional data structures [48]. Tries for parameterized types
like lists or binary trees are represented as Standard ML functors. While this
approach works for regular data types, it fails for nested data types such as
Sequ. In the latter case data types of second-order kind are indispensable. The
material in this section has been taken from Hinze [22].

Exercise 1. (Simple exercise to experiment with Generic Haskell.) Define func-
tion depth, which returns the depth of a value. The depth of a value is the
maximum number of constructors encountered on a path from the root to a leaf.
For example, given the data type Tree:

data Tree a = Leaf a | Node (Tree a) a (Tree a)

the depth of

exTree = Node (Leaf 1) 2 (Node (Node (Leaf 6) 0 (Leaf (−11))) 4 (Leaf 4))

is 4.

Exercise 2. (More difficult exercise about dictionaries.) Define function insert as
a type-indexed function with a kind-indexed kind. You can download the code for
the functions described in this section from http://www.generic-haskell.org,
including the solution to this exercise. You might want to avoid looking at the
implementation of insert while solving this exercise.

3 Generic Programming for XML tools

An XML document is usually structured according to a document type definition
(DTD). DTDs are another formalism for specifying data types (or grammars,
abstract syntax). This section shows how an XML compressor is implemented
as a generic program, and it discusses which other classes of XML tools would
profit from an implementation as a generic program. The example shows how
generic programming can be used to implement XML tools such as XML editors,
databases, and compressors, that depend on the DTD of an input XML docu-
ment. The resulting tools usually perform better because knowledge of the DTD
can be used to optimise the tools, and are smaller, because all DTD handling is
dealt with in the generic programming compiler.

The Generic Haskell code for this section can be downloaded from the ap-
plications page on http://www.generic-haskell.org/. It is also distributed
together with the Generic Haskell compiler.

3.1 Introduction

XML Tools. Since W3C released XML [55], the de facto data format standard on
the web, hundreds of XML tools have been developed. There exist XML editors,
XML databases, XML converters, XML parsers, XML validators, XML search
engines, XML encryptors, etc. Information about XML tools is available from
many sites, see for example [18, 20]. Flynn’s book [17] provides a description of
some older tools.

Usage of DTDs in XML Tools. An XML document is usually structured accord-
ing to a Document Type Definition (DTD) or a schema. An XML document is
valid with respect to a DTD if it is structured according to the rules (elements)
specified in the DTD. So a validator is a tool that critically depends on a DTD.
Some other classes of tools, such as the class of XML editors, also critically de-
pend on the presence of a DTD. An XML editor can only support editing of
an XML document well, for example by suggesting possible children or listing
attributes of an element, if it knows about the element structure and attributes
of elements. These classes of tools depend on a DTD, and do the same thing
(modulo structure differences) for different DTDs. We claim that many classes
of XML tools are generic programs, or would benefit from being viewed as generic
programs. We call such tools DTD-indexed XML tools.

Generic Programming for XML Tools. Since DTD-indexed XML tools are generic
programs, it should help to implement such tools as generic programs. Imple-
menting an XML tool as a generic program has several advantages:

– Development time. Generic programming supports the construction of type-
(or DTD-) indexed programs. So all processing of DTDs and programs de-
fined on DTDs can be left to the compiler, and does not have to be im-
plemented by the tool developer. Furthermore, the existsing library of often
used basic generic programs, for example, for comparing, encoding, etc., can
be used in generic programs for XML tools.

– Correctness. An instance of a typable generic program is typeable. This im-
plies that valid documents will be transformed to valid documents, possibly
structured according to another DTD. Thus generic programming supports
constructing type correct XML tools.

– Efficiency. The generic programming compiler may perform all kinds of op-
timisations on the code, such as deforestation, partial evaluation or fusion,
which are difficult to conceive or implement by an XML tool developer.

This section discusses which classes of XML tools are DTD indexed, and how
they can be implemented as generic programs.

Generic programming can also be used for XML tools that are not DTD
indexed, but then most of the above advantages no longer apply.

3.2 XML compressors

Compression for XML documents. XML documents may become (very) large
because of the markup that is added to the content. Because of the repetitive
structure of many XML documents, these documents can be compressed by quite
a large factor.

Existing XML compressors. We know of four XML compressors2:

– XMLZip [12]. XMLzip cuts its argument XML file (viewed as a tree) at
a certain depth, and compresses the upper part separately from the lower
part, both using a variant of zip or LZW [59]. This allows fast access to doc-
uments, but results in worse compression ratios compared with the following
compressors.

– XMill [40]. XMill is a compressor that separates the structure of the XML
document from the contents, and compresses structure and contents sep-
arately. Furthermore, it groups related data items (such as dates), and it
applies semantic compressors to data items with a particular structure.

2 Actually, after we wrote this paragraph, we found two more XML compressors. This
field seems to develop fast. The final version of these lecture notes will contain the
references.

– ICT’s XML-Xpress [30] is a commercial compression system for XML files
that uses ‘Schema model files’ to provide support for files conforming to a
specific XML schema. The basic idea of this system is the same as the idea
underlying the compressor we will describe below.

– Millau [19] is a system for efficient encoding and streaming of XML struc-
tures. It also separates structure and content, and uses the associated schema
(if present) for compressing the structure.

XML compression and DTDs. XML compressors are DTD indexed. For example,
consider the following small XML file:

<book lang="English">
<title> Dead Famous </title>
<author> Ben Elton </author>
<date> 2001 </date>
</book>

This file may be compressed by separating the structure from the data, and
compressing the two parts separately. For compressing the structure we can
make good use of the DTD. If we know how many elements, say n, appear in
the DTD (the DTD for the above document contains at least 4 elements), we
can replace each occurrence of the markup of an element in an XML file which
is valid with respect to the DTD by log2 n bits. This simple idea is the main
idea behind the following tool, and has been described in the context of data
conversion by Jansson and Jeuring [31, 35].

3.3 Implementing an XML compressor as a generic program

We have implemented an XML compressor, called XComprez, as a generic pro-
gram. XComprez separates structure from contents, compresses the structure
using knowledge about the DTD, and compresses the contents using a variant
of zip [59]. Thus we replace each element, or rather, the pair of open and close
keywords of the element, by the minimal number of bits required for the element
given the DTD. We distinguish four components in the tool:

– a component that translates a DTD to a data type,
– a component that separates a value of any data type into its structure and

its contents,
– a component that encodes the structure replacing constructors by bits,
– and a component for compressing the contents.

Of course, we have also implemented a decompressor, but since it is dual, hence
very similar, to the compressor, we omit its description. See the website for
XComprez [37] for the latest developments on XComprez. The Generic Haskell
source code for XComprez can be obtained from the website.

Translating a DTD to a data type. A DTD can be translated to one or more
Haskell data types. For example, the following DTD:

<!ELEMENT book (title,author,date,(chapter)*)>
<!ELEMENT title (#PCDATA)>
<!ELEMENT author (#PCDATA)>
<!ELEMENT date (#PCDATA)>
<!ELEMENT chapter (#PCDATA)>
<!ATTLIST book lang (English | Dutch) #REQUIRED>

can be translated to the following data types:

data Book = Book Book Attrs Title Author Date [Chapter ]
data Book Attrs = Book Attrs{bookLang :: Lang }
data Lang = English | Dutch
newtype Title = Title String
newtype Author = Author String
newtype Date = Date String
newtype Chapter = Chapter String

We have used the Haskell library HaXml [58], in particular the functionality in
the module DtdToHaskell to obtain a data type from a DTD, together with func-
tions for reading (parsing) and writing (pretty printing) valid XML documents
to and from a value of the generated data type. For example, the following value
of the above DTD:

<book lang="English">
<title> Dead Famous </title>
<author> Ben Elton </author>
<date> 2001 </date>
<chapter>Introduction </chapter>
<chapter>Preliminaries</chapter>
</book>

is translated to the following value of the data type Book :

Book Book Attrs{bookLang = English }
(Title "  Dead Famous  ")
(Author " Ben Elton    ")
(Date "   2001         ")
[Chapter "Introduction "
,Chapter "Preliminaries"
]

An element is translated to a value of a data type using just constructors and
no labelled fields. An attribute is translated to a value that contains a labelled
field for the attribute. Thus we can use the Generic Haskell constructs Con and
Label to distinguish between elements and attributes in generic programs.

Separating structure and contents. The contents of an XML document is ob-
tained by extracting all PCData and all CData from the document. In Generic
Haskell, the contents of a value of a data type is obtained by extracting all strings
from the value. For the above example value, we obtain the following result:

["  Dead Famous  "
, " Ben Elton    "
, "   2001         "
, "Introduction "
, "Preliminaries"
]

The generic function extract , which extracts all strings from a value of a data
type, is defined as follows:

type Extract〈〈?〉〉 t = t → [String ]
type Extract〈〈κ→ ν〉〉 t = ∀a .Extract〈〈κ〉〉 a → Extract〈〈ν〉〉 (t a)
extract〈t :: κ〉 :: Extract〈〈κ〉〉 t
extract〈Unit〉 Unit = [ ]
extract〈String〉 s = [s ]
extract〈:+:〉 eA eB (Inl x ) = eA x
extract〈:+:〉 eA eB (Inr y) = eB y
extract〈:*:〉 eA eB (x :*: y) = eA x ++ eB y
extract〈Con c〉 e (Con b) = e b
extract〈Label l〉 e (Label b) = e b

Note that it is possible to give special instances of a type-indexed function on
a particular type, as with extract〈String〉 in the above definition. Furthermore,
because DtdToHaskell translates any DTD to a data type of kind ?, we could
have defined extract just on data types of kind ?. However, higher-order kinds
pose no problems. Finally, note that the operator ++ in the product case is a
source of inefficiency. It can be removed using the standard lifting to the function
level approach.

The structure from an XML document is obtained by removing all PCData
and CData from the document. In Generic Haskell, the structure, or shape, of a
value of a data type is obtained by replacing all strings by units (empty tuples).
Thus we obtain a value of a new data type, in which occurrences of the type
String have been replaced by the type (). This is another example of a type-
indexed data type [24]. For example, the type we obtain from the data type

Book is isomorphic to the following data type:

data SHAPEBook = SHAPEBook SHAPEBook Attrs
SHAPETitle
SHAPEAuthor
SHAPEDate
[SHAPEChapter ]

data SHAPEBook Attrs = SHAPEBook Attrs{bookLang :: SHAPELang }
data SHAPELang = SHAPEEnglish | SHAPEDutch
newtype SHAPETitle = SHAPETitle ()
newtype SHAPEAuthor = SHAPEAuthor ()
newtype SHAPEDate = SHAPEDate ()
newtype SHAPEChapter = SHAPEChapter ()

and the structure of the example value is

shapeBook = SHAPEBook ( SHAPEBook Attrs{bookLang = SHAPEEnglish })
( SHAPETitle ())
( SHAPEAuthor ())
( SHAPEDate ())
[ SHAPEChapter ()
, SHAPEChapter ()
]

The type-indexed data type SHAPE replaces occurrences of String in a data
type by Unit .

type SHAPE 〈Unit〉 = SH1 Unit
type SHAPE 〈String〉 = SHString Unit
type SHAPE 〈:+:〉 sa sb = SHEither (Sum sa sb)
type SHAPE 〈:*:〉 sa sb = SHProd (Prod sa sb)
type SHAPE 〈Con〉 sa = SHCon (Con sa)
type SHAPE 〈Label〉 sa = SHLabel (Label sa)

The generic function shape returns the shape of a value of any data type, using
the constructors of the type-indexed data type SHAPE .

type Shape〈〈?〉〉 t = t → SHAPE 〈t〉
type Shape〈〈κ→ ν〉〉 t = ∀a .Shape〈〈κ〉〉 a → Shape〈〈ν〉〉 (t a)

shape〈t :: κ〉 :: Shape〈〈κ〉〉 t
shape〈Unit〉 u = SH1 Unit
shape〈String〉 s = SHString Unit
shape〈:+:〉 sa sb (Inl a) = SHEither (Inl (sa a))
shape〈:+:〉 sa sb (Inr b) = SHEither (Inr (sb b))
shape〈:*:〉 sa sb (a :*: b) = SHProd ((sa a) :*:(sb b))
shape〈Con c〉 sa (Con b) = SHCon (Con (sa b))
shape〈Label l〉 sa (Label b) = SHLabel (Label (sa b))

Given the shape and the contents (obtained by means of function extract) of a
value we obtain the original value by means of function insert :

insert〈t :: ?〉 :: SHAPE 〈t〉 → [String ]→ t

The type-indexed definition (with a kind-indexed type) of function insert is
omitted.

Encoding constructors. A constructor of a value of a data type is encoded as
follows. First calculate the number n of constructors of the data type. Then
calculate the position of the constructor in the list of constructors of the data
type. Finally, replace the constructor by the bit representation of its position,
using log2 n bits. For example, in a data type with 6 constructors, the third
constructor is encoded by 010. Note that we start counting with 0. Furthermore,
note that a value of a data type with a single constructor is represented using
0 bits. So the values of all types except for String and Lang in the example are
represented using 0 bits.

All constructor descriptions of a data type can be obtained by means of
function constructors from the module Collect, which can be found in the library
of Generic Haskell.

constructors〈t :: ?〉 :: [ConDescr ]

Function constructors is defined for arbitrary kinds in module Collect. The
generic function encode takes a shape value, and encodes it. Since it needs the
constructors of a data type to encode a shape value, function encode depends
on function constructors.

dependency encode ← constructors encode

On types of kind ?, encode takes maybe a list of constructor descriptions (the
constructor descriptions currently in scope: note that this list may change when
traversing a value of a data type that refers to other data types) and a shape
value, and returns a list of bits.

type Encode〈〈?〉〉 t = Maybe [ConDescr ]→ SHAPE 〈t〉 → [Bit ]
type Encode〈〈κ→ ν〉〉 t =
∀u .Collect0 〈〈κ〉〉 [ConDescr ]→ Encode〈〈κ〉〉 u → Encode〈〈ν〉〉 (t u)

type Collect0 〈〈?〉〉 a = a
type Collect0 〈〈κ→ ν〉〉 a = ∀u .Collect0 〈〈κ〉〉 a → Collect0 〈〈ν〉〉 a

The only interesting cases in the definition of function encode are the sum and
constructor case. We first give the uninteresting cases:

encode〈t :: κ〉 :: Encode〈〈κ〉〉 t
encode〈Unit〉 = λ → [ ]
encode〈String〉 = λ → [ ]
encode〈:*:〉 cA eA cB eB =

λ (SHProd (a :*: b))→ eA Nothing a ++ eB Nothing b
encode〈Label l〉 cA eA = λ (SHLabel (Label a))→ eA Nothing a

For Unit and String there is nothing to encode. Note that the product case takes
four arguments, which correspond to the constructors and encoding of the left
and right argument of the product, respectively. The encoding functions for the
arguments are called with no constructor descriptions (Nothing), since whenever
a new constructor is encountered in a product, it might be from another data
type, and the constructors have to be recalculated.

In the sum case we calculate the constructors, if necessary (implemented
by maybe), and encode the arguments of sum with the constructors. The en-
coding happens in the constructor case of function encode. We use function
intinrange2bits to calculate the bits for the position of the argument construc-
tor in the constructor list, given the number of constructors of the data type
currently in scope. The definition of intinrange2bits is omitted.

encode〈:+:〉 cA eA cB eB =
let cR = cA ++ cB
in λcs → let cs ′ = maybe cR id cs

in eA (Just cs ′) ‘shapejunc‘ eB (Just cs ′)
encode〈Con c〉 cA eA =

λcs (SHCon (Con a))→
let cs ′ = maybe [c ] id cs
in intinrange2bits (length cs ′) (fromJust (elemIndex c cs ′))

++ eA Nothing a

shapejunc :: (a → c)→ (b → c)→ (SHAPE 〈:+:〉 a b)→ c
shapejunc f g (SHEither (Inl x )) = f x
shapejunc f g (SHEither (Inr x )) = g x
intinrange2bits :: Int → Int → [Bit ]

We omit the definitions of the functions to decode a list of bits into a value of
a data type. These functions are the inverses of the functions defined in this
section.

Compressing the contents. Finally, the contents of an XML document have to be
compressed. At the moment we use zip to compress the strings obtained from the
document. In the future, we envisage more sophisticated compression methods
for the contents, similar to the methods used in XMill.

Huffman coding. A relatively simple way to improve XComprez it is to analyze
some source files that are valid with respect to the DTD, count the number of
occurrences of the different elements (constructors), and apply Huffman coding.
We have implemented this rather simple extension [37].

Analysis. How does the compressor described in the previous subsection compare
with the existing XML compressors? The following analysis is limited, because
we have not been able to obtain the executables or the source code of some of
the existing compressors. Since the goal of XMLZip is different from our and
the other compressors goal (fast access to compressed documents), we do not
compare with XMLZip.

Compression ratio. XML-Xpress has been tested extensively against XMill, and
achieves compression results that are about 80% better than XMill. We have
performed some initial tests comparing XComprez and XMill. The tests are not
representative, and it is impossible to draw hard conclusions from the results.
However, on our test examples XComprez is 40% to 50% better than XMill.
We think this improvement in compression ratio is considerable. As a schema
contains more information about an XML document than a DTD, it is not
surprising that our compressor does not achieve the same compression ratios
as XML-Xpress. However, when we replace HaXml by a tool that generates
a data type for a schema, we expect that we can achieve similar compression
ratios as XML-Xpress. We have not been able to test against Millau, but from
its description we expect that Millau achieves compression ratios that are a bit
worse than the compression ratios achieved by XComprez, as Millau uses a
fixed number of bits for some elements or attributes, independent of the DTD
or Schema.

Code size. With respect to code size, the difference between XMill and XCom-
prez is dramatic: XMill is written in almost 20k lines of C++. The main func-
tionality of XComprez is less than 300 lines of Generic Haskell code. Of course,
for a fair comparison we have to add some of the HaXml code (which is a library
distributed together with almost all compiler and interpreters for Haskell), the
code for handling bits, and the code for implementing the as yet unimplemented
features of XMill. We expect to be able implement all of XMill’s features in
about 20% of the code size of XMill. We have not been able to obtain the source
code of the (commercial) XML-Xpress.

3.4 DTD-indexed XML Tools

This section discusses whether or not several classes of XML tools are DTD
indexed. We briefly introduce each of the classes of tools, and we discuss whether
the class is DTD indexed and whether the available tools make use of this fact.
Furthermore, whenever applicable, we discuss where HaXml and Generic Haskell
might help in implementing an XML tool. Note that some (classes of) XML tools
develop very fast, and that some of the information given in this section may be
out of date.

We will discuss the following classes of tools:

– XML converters, parsers, and validators
– XML databases and search tools
– XML editors
– XML encryptors
– XML publishing tools
– XML version management tools

The class of XML compressors does not appear in this list, but has been discussed
in the previous section. This is not a complete list of classes of XML tools, but
this list includes many of the XML tools in use today.

XML converters, parsers, and validators. Since they are very similar, we discuss
the classes of XML converters, parsers, and validators together.

For an overview of XML parsers, see [18]. There are several variants of XML
parsers. Most XML parsers parse an arbitrary XML document to a universal
tree (a DOM). DTDs play no role when parsing: validity of an XML document
with respect to a DTD is checked in a separate phase, for example by an XML
validator. These parsers are not DTD indexed.

Using Generic Haskell to develop an XML parser we would obtain a tool that
takes a DTD as argument, and returns a parser for documents of the argument
DTD. Thus the parser is automatically a validator. Any element that would turn
the document into an invalid document would lead to a parse error. The HaXml
library, in particular the module DtdToHaskell, contains a generic parser of this
kind. Since this technology lies at the basis of our tools, we want to reimplement
the read and show functions from DtdToHaskell in Generic Haskell, and add a
module SchemaToHaskell.

For an overview of XML converters, see [18]. There exist two classes of XML
converters: XML to XML converters, and non-XML to XML converters. For both
of these classes there exist specific and generic (that is DTD-indexed) tools. An
example of a specific non-XML to XML converter is RTF2XML [28]. Examples
of generic non-XML to XML converters are Some2xml and Jedi [53, 26]. In both
Some2xml and Jedi it is possible to specify patterns that are to be mapped on
XML. If it were also possible to specify the result DTD, we would obtain a generic
parser. At the moment Generic Haskell is of little use here, but future versions of
Generic Haskell might offer functionality that is useful for implementing generic
converters. The converters from XML to another format are discussed in the
XML publishing tools section.

XML databases and search tools. XML documents can be searched by means
of queries. Since XML query languages also play an important role in XML
databases, we discuss XML databases and search tools under a common heading.

XML databases are used to store XML documents in a database. There are
several ways to store an XML document in a database, but each of these can be
classified as either structured or unstructured. The XML databases that store
documents in a structured way are DTD-indexed XML tools. The DTD is used
to determine the tables in the database, and may be used to optimise queries
etc. Being DTD indexed can be of great help when searching or querying XML
documents: indexes can be built based on DTDs, subtrees can be skipped when
searching, etc. According to Abiteboul et al [1] current databases are not DTD
indexed. However, the field of XML databases is developing fast, and we expect
that there may already be DTD-indexed XML databases. Since most of these
tools are commercial tools, see for example [11], it is difficult to check whether
or not they are DTD indexed, and to compare implementations.

XML editors. An XML editor supports editing an XML document. Most XML
editors support viewing XML documents in different ways, and they suggest
elements and attributes that may be inserted at a given position. There are too

many XML editors to list here. An incomplete list of XML editors can be found
at the Proxima site [38]. An XML editor is a nice example of a DTD-indexed
XML tool. Most XML editors are DTD indexed, and we think they should be.
We have started on the core of a generic editor [14], but a lot of work remains in
order to obtain a full-fledged XML editor. Again, as most of the existing XML
editors are commercial tools, see for example [52, 2], it is difficult to compare
implementations.

XML encryptors. An XML encryptor encrypts an XML document. Since the
encrypted document gives nothing away of the structure of the input document,
we see no application for generic programming here. Indeed, encryption would
be weaker if it were based on the structure of a document.

XML publishing tools. An XML publishing tool, like for example Cocoon [3],
takes an XML document and a target type on which to publish the document,
and maybe a style sheet for this type, and returns a document which can be
published. The tool traverses the input document, using the style sheet. Existing
publishing tools are not DTD indexed. DTD indexing might help in constructing
a publishing tool, since knowledge about the DTD can be used to optimise the
traversal.

XML version management tools. IBM has developed a tool called treediff [29]
which compares two XML files and points out the differences between the two
files. A similar tool has been developed by Dommit [15]. We think that it would
not be difficult to implement such a tool in Generic Haskell. Chawathe [5] has
developed algorithms for comparing hierarchally structured data (such as XML
documents). It is easy to implement the minimum-cost edit distance algorithm
given by Chawathe as a generic program, by printing values to the format ex-
pected by the algorithm, and parsing, to a value of the original type, the tree
obtained by applying the minimum cost edit script to the printed argument.
Types do not play an essential role in this algorithm.

3.5 Related work

Most XML tools are built using the DOM or the SAX for manipulating XML
documents. Using the DOM or the SAX usually implies that an XML document
does not have a type. It follows that these standards are not a lot of help when
developing tools that critically depend on the type (DTD) of a document.

There are a number of XML-specific (query) languages, such as for exam-
ple XDuce [25], XMλ [47, 51], XSLT [56], XML query algebras [16], Yatl [9].
In many of these languages, XML documents are native values. Each of these
languages has a number of features, such as regular expression pattern match-
ing, type inference, or regular expression types, that support the construction of
programs that manipulate XML documents. However, none of these languages
have features that support the construction of DTD-indexed XML tools. We ex-
pect that extending XMλ with a construct that supports defining DTD-indexed

functions would result in a very useful language for developing XML tools, but
unfortunately an implementation of XMλ does not seem to exist.

3.6 Conclusions

We have shown that the combination of HaXml and generic programming as in
Generic Haskell is very useful for implementing DTD-indexed XML tools. Using
generic programming, such tools become easier to write, because a lot of the
code pertaining to DTD handling and optimisation is obtained from the generic
programming compiler, and the resulting tools are more effective, because they
directly depend on the DTD. For example, DTD-indexed XML compressors, such
as XComprez described in this paper, compress considerably better than XML
compressors that don’t take the DTD into account, such as XMill. Furthermore,
our compressor is much smaller than XMill.

It remains to develop other DTD-indexed XML tools, and a library that
supports the development of XML tools using Generic Haskell. We have started
on an XML editor in Generic Haskell, see [14], but a lot of work remains to be
done. However, we hope to (further) develop at least our XML compressor, an
XML editor, part of an XML version management tool, and an XML database
this year.

Although we think Generic Haskell is very useful for developing DTD-indexed
XML tools, there are some features of XML tools that are harder to express in
Generic Haskell. Some of the functionality in the DOM, such as the methods
childNodes and firstChild in the Node interface, is hard to express in a typed
way. Flexible extensions of type-indexed data types [24] might offer a solution to
this problem. We think fusing HaXml, or a tool based on Schemas, with Generic
Haskell, obtaining a ‘domain-specific’ language [6] for generic programming on
DTDs or Schemas is a promising approach.

For tools that do not depend on a DTD we can use the untyped approach
from HaXml to obtain a tool that works for any document. However, most of
the advantages of Generic Programming no longer apply.

Exercise 3. (Easy exercise about XComprez.) In order to implement Huffman
coding for XComprez, we have to analyse representative documents of a data
type. So we want to count the constructors that appear in a value of a data type.
For example,

data Tree = Leaf Int | Node Tree Int Tree
?countCon (Node (Leaf 1) 3 (Node (Leaf 2) 1 (Leaf 5)))
[("Leaf", 3), ("Node", 2)]
data List = Nil | Cons Char List
?countCon (Cons 2 (Cons 3 (Cons 6 Nil)))
[("Nil", 1), ("Cons", 3)]

Define the type-indexed function countCon, together with its kind-indexed type.

Exercise 4. (More involved exercise about adapting a generic program.) Adapt
the current version of XComprez such that it can use Huffman coding instead
of the standard constructor encoding used in this section. Make sure also other
encodings can be used. Solutions to these exercises can be found on the webpage
for XComprez.

4 The zipper

This section shows how to define a zipper for an arbitrary data type. This is
an advanced example demonstrating the full power of a type-indexed data type
together with a number of type-indexed functions working on it.

The zipper is a data structure that is used to represent a tree together with a
subtree that is the focus of attention, where that focus may move left, right, up
or down in the tree. The zipper is used in tools where a user interactively manip-
ulates trees, for instance, in editors for structured documents such as proofs and
programs. For the following it is important to note that the focus of the zipper
may only move to recursive components. Consider as an example the data type
Tree:

data Tree a = Empty | Node (Tree a) a (Tree a).

If the left subtree of a Node constructor is selected, moving right means moving
to the right tree, not to the label of type a. This implies that recursive positions
in trees play an important rôle in the definition of a generic zipper data structure.
To obtain access to these recursive positions, we have to be explicit about the
fixed points in data type definitions. The zipper data structure is then defined
by induction on the so-called pattern functor of a data type.

The tools in which the zipper is used, allow the user to repeatedly apply
navigation or edit commands, and to update the focus accordingly. In this section
we define a type-indexed data type for locations, which consist of a subtree (the
focus) together with a context, and we define several navigation functions on
locations.

The Generic Haskell code for this section can be downloaded from the ap-
plications page on http://www.generic-haskell.org/. It is also distributed
together with the Generic Haskell compiler.

4.1 Preliminaries

Data types as fixed points. As mentioned above, in order to use the zipper, we
have to be explicit about the fixed points in data type definitions.

newtype Fix f = In{out :: f (Fix f )}

For example, the data types of natural numbers and trees can be defined as fixed
points as follows:

data NatF a = ZeroF | SuccF a
type Nat = Fix NatF
data TreeF a = LeafF Char | ForkF a a
type Tree = Fix TreeF

It is easy to convert between data types as fixed points and the original data type
definitions of natural numbers and trees. Note that nested data types and mutu-
ally recursive data types cannot be defined in terms of this particular definition
of Fix .

The identity type-indexed data type. In the following subsections we will fre-
quently use the identity type-indexed data type Gid . The definition of Gid is
omitted. We assume there exist functions mkid and unid , which turn a value of
type t into a value of type Gid〈t〉 and vice versa:

mkid〈t :: κ〉 :: MkId〈〈κ〉〉 t
type MkId〈〈?〉〉 t = t → Gid〈t〉
type MkId〈〈κ→ ν〉〉 t = ∀u .MkId〈〈κ〉〉 u → MkId〈〈ν〉〉 (t u)
unid〈t :: κ〉 :: UnId〈〈κ〉〉 t
type UnId〈〈?〉〉 t = Gid〈t〉 → t
type UnId〈〈κ→ ν〉〉 t = ∀u .UnId〈〈κ〉〉 u → UnId〈〈ν〉〉 (t u)

Function mkid and unid are each others inverse.

Lifted Maybe. The lifted Maybe type is called LMaybe, and is defined by:

data LMaybe f a = LNothing | LJust (f a)

Functions unlift :: LMaybe f a → Maybe (f a) and lift :: Maybe (f a) →
LMaybe f a convert between the two types. We will use the functions out
and in

out = unlift . out
in = In . lift

4.2 Locations

A location is a subtree, together with a context, which encodes the path from
the top of the original tree to the selected subtree. The type-indexed data type
Loc returns a type for locations given an argument pattern functor.

Loc〈f :: ?→ ?〉 :: ?
type Loc〈f 〉 = (Fix f ,Context〈f 〉 (Fix f ))
Context〈f :: ?→ ?〉 :: ?→ ?
type Context〈f 〉 r = Fix (LMaybe (Ctx 〈f 〉 r)).

The type Loc is defined in terms Context , which constructs the context param-
eterized by the original tree type. Note that we use generic abstraction on types
here, instead of on generic functions. This feature has not been added to Generic
Haskell yet, so in the actual Generic Haskell code, Loc and Context are replaced
by their right-hand sides. The Context of a value is either empty (represented
by LNothing in the LMaybe type), or it is a path from the root down into the
tree. Such a path is constructed by means of the argument type of LMaybe:
the type-indexed data type Ctx . The type-indexed data type Ctx is defined by
induction on the pattern functor f of the original data type. It can be seen as
the derivative (as in calculus) of the type f . If the derivative of f is denoted by
f ′, we have

const ′ = 0
(f + g)′ = f ′ + g ′

(f × g)′ = f ′ × g + f × g ′

It follows that Ctx depends on the identity type-indexed data type Gid . Depen-
dencies on type-indexed data types work in the same way as dependencies on
type-indexed functions. The identity type-indexed data type is only used in the
product case for Ctx .

dependency Ctx ← Gid Ctx
Ctx 〈f :: ?→ ?〉 :: ?→ ?→ ?
type Ctx 〈Unit〉 = CTXUnit 0
type Ctx 〈Int〉 = CTXInt 0
type Ctx 〈Char〉 = CTXChar 0
type Ctx 〈:+:〉 iA cA iB cB = CTXSum (Sum cA cB)
type Ctx 〈:*:〉 iA cA iB cB = CTXProd (Sum (Prod cA iB) (Prod iA cB))
type Ctx 〈Con〉 iA cA = CTXCon cA
type Ctx 〈Label〉 iA cA = CTXLab cA

This definition can be understood as follows. Since it is not possible to descend
into a constant, the constant cases do not contribute to the result type, which is
denoted by the ‘empty type’ 0. Descending in a value of a sum type follows the
structure of the input value. Finally, there are two ways to descend in a product:
descending left, adding the contents to the right of the node to the context, or
descending right, adding the contents to the left of the node to the context.

For example, for natural numbers and trees we obtain ismorphic versions of
the following context types:

type ContextNat r = Fix (LMaybe (NatC r))
type ContextTree r = Fix (LMaybe (TreeC r))
data NatC r a = ZeroC 0 | SuccF a
data TreeC r a = LeafC 0 | NodeF (a, r) (r , a)

Note that if we assume 0 is a unit of + the context of a natural number is
isomorphic to a natural number (the context of m in n is n − m), and the
context of a Tree applied to the data type Tree itself is isomorphic to the type
Ctx Tree introduced in Section 1.

McBride [43] also defines a type-indexed zipper data type. His zipper slightly
deviates from Huet’s and our zipper: the navigation functions on McBride’s
zipper are not constant time anymore. The observation that the context of a
data type is its derivative is due to McBride.

4.3 Navigation functions

We define type-indexed functions on the type-indexed data types Loc, Context ,
and Ctx for navigating through a tree. All of these functions act on locations.
These are the basic functions for the zipper.

Function down. The function down is a type-indexed function that moves down
to the leftmost recursive child of the current node, if such a child exists. Other-
wise, if the current node is a leaf node, then down returns the location unchanged.
The instantiation of down to the data type Tree has been given in Section 1.
The function down satisfies the following property:

∀l . down〈f 〉 l 6= l =⇒ (up〈f 〉 · down〈f 〉) l = l ,

where function up goes up in a tree. So first going down the tree and then up
again is the identity function on locations in which it is possible to go down.

Since down moves down to the leftmost recursive child of the current node,
the inverse equality down〈f 〉 · up〈f 〉 = id does not hold in general. However,
there does exist a natural number n such that

∀l . up〈f 〉 l 6= l =⇒ (right〈f 〉n · down〈f 〉 · up〈f 〉) l = l .

The function down is defined as follows.

down〈f :: ?→ ?〉 :: Loc〈f 〉 → Loc〈f 〉
down〈f 〉 (t , c) = case first〈f 〉 (out t) c of

Just (t ′, c′)→ (t ′, in (Just c′))
Nothing → (t , c).

The helper function first is a type-indexed function that possibly returns the
leftmost recursive child of a node, together with the context (a value of type
Ctx 〈f 〉 r (Fix f )) of the selected child. The function down then turns this
context into a value of type Context by inserting it in the right (‘non-top’)
component of a sum by means of Just , and applying the fixed point constructor
in to it.

The value out t is of type f (Fix f ). We want to obtain the leftmost oc-
currence of type Fix f in out t . For this purpose we define first as a generic
abstraction of a function first ′.

first〈f :: ?→ ?〉 :: f (Fix f )→ c → Maybe (Fix f ,Ctx 〈f 〉 (Fix f ) c)
first〈f 〉 x c = first ′ (f ) first ′Rec id x c

The first argument of first ′ (f ), function first ′Rec, is the function that is applied
to the values of type Fix f , when x :: f (Fix f ). Since we want to return a value

of type Fix f , together with its context, function first ′Rec is the curried version
of Just :

first ′Rec t c = Just (t , c)

Function first ′ does the real work. It depends on the identity function mkid .

dependency first ′ ← first ′ mkid
first ′ (t :: κ) :: ∀a c .First〈〈κ〉〉 t a c
type First〈〈?〉〉 t a c = t → c → Maybe (a,Ctx 〈t〉)
type First〈〈κ→ ν〉〉 t a c = ∀u .First〈〈κ〉〉 u a c → MkId〈〈κ〉〉 u → First〈〈l〉〉 (t u) a c

Because of the dependency, first ′ is also applied to the identity function in the
generic abstraction above.

first ′〈Unit〉 t c = Nothing
first ′〈Int〉 t c = Nothing
first ′〈Char〉 t c = Nothing
first ′〈:+:〉 fA mA fB mB (Inl x ) c = do (t , cx )← fA x c

return (t ,CTXSum (Inl cx ))
first ′〈:+:〉 fA mA fB mB (Inr y) c = do (t , cy)← fB y c

return (t ,CTXSum (Inr cy))
first ′〈:*:〉 fA mA fB mB (x :*: y) c = (do (t , cx )← fA x c

return (t ,CTXProd (Inl (cx :*:mB y)))
)
‘mplus‘
(do (t , cy)← fB y c

return (t ,CTXProd (Inr (mA x :*: cy)))
)

first ′〈Con d〉 fA mA (Con t) c = do (t , cx )← fA t c
return (t ,CTXCon cx )

Here, return is obtained from the Maybe monad, and mplus is the standard
monadic plus, given by

mplus :: Maybe a → Maybe a → Maybe a
Nothing ‘mplus‘ m = m
Just a ‘mplus‘ m = Just a.

The function first returns the value and the context at the leftmost recursive
position. So in the product case, it first tries the left component, and only if it
fails, it tries the right component.

The definitions of functions up, right and left are not as simple as the defini-
tion of down, since they are defined by pattern matching on the context instead
of on the tree itself. We will just define functions up and right , and leave function
left to the reader.

Function up. The function up moves up to the parent of the current node, if the
current node is not the top node.

up〈f :: ?→ ?〉 :: Loc〈f 〉 → Loc〈f 〉
up〈f 〉 (t , c) = case out c of

Nothing → (t , c)
Just c′ → fromJust$

do {ft ← insert〈f 〉 c′ t ;
c′′ ← extract〈f 〉 c′;
return (In ft , c′′)}.

Remember that Nothing denotes the empty top context. The navigation function
up uses two helper functions: insert and extract . The latter returns the context
of the parent of the current node. Note that each element of type Ctx 〈f 〉 t c
has at most one c component (by an easy inductive argument), which marks
the context of the parent of the current node. The polytypic function extract
extracts this context. Just as function first , function extract is defined as a
generic abstraction of function extract ′.

extract〈f :: ?→ ?〉 :: Ctx 〈f 〉 t c → Maybe c
extract〈f 〉 c = extract ′〈f 〉 extract ′Rec c
extract ′Rec c = Just c
extract ′〈t :: κ〉 :: ∀a .Extract〈〈κ〉〉 t a
type Extract〈〈?〉〉 t a = Ctx 〈t〉 → Maybe a
type Extract〈〈κ→ ν〉〉 t a = ∀u .Extract〈〈κ〉〉 u a → Extract〈〈ν〉〉 (t u) a

Note that extract is polymorphic in t and in c. Function extract ′ is a simple
function that traverses a context value.

extract ′〈Unit〉 c = Nothing
extract ′〈Int〉 c = Nothing
extract ′〈Char〉 c = Nothing
extract ′〈:+:〉 eA eB (CTXSum (Inl cx )) = eA cx
extract ′〈:+:〉 eA eB (CTXSum (Inr cy)) = eB cy
extract ′〈:*:〉 eA eB (CTXProd (Inl (cx :*: y))) = eA cx
extract ′〈:*:〉 eA eB (CTXProd (Inr (x :*: cy))) = eB cy
extract ′〈Con c〉 eA (CTXCon cx ) = eA cx
extract ′〈Label l〉 eA (CTXLab cx ) = eA cx

Function insert takes a context and a tree, and inserts the tree in the current
focus of the context, effectively turning a context into a tree. To obtain such
a tree, we have to remove the occurrences of constructors of the identity type-
indexed data type, for which we use function unid .

dependency insert ′ ← insert ′ unid

Function insert is defined in a similar fashion as extract and first . We first give
a generic abstraction:

insert〈f :: ?→ ?〉 :: Ctx 〈f 〉 (Fix f ) c → Fix f → Maybe (f (Fix f ))
insert〈f 〉 c t = insert ′〈f 〉 insert ′Rec id c

where insert ′Rec c = Just t
insert ′〈t :: κ〉 :: Insert〈〈κ〉〉 t
type Insert〈〈?〉〉 t = Ctx 〈t〉 → Maybe t
type Insert〈〈κ→ ν〉〉 t = ∀u . Insert〈〈κ〉〉 u → UnId〈〈κ〉〉 u → Insert〈〈ν〉〉 (t u)

and then define the type-indexed function insert ′

insert ′〈Unit〉 c = Nothing
insert ′〈Int〉 c = Nothing
insert ′〈Char〉 c = Nothing
insert ′〈:+:〉 iA uA iB uB (CTXSum (Inl cx )) = do x ← iA cx

return (Inl x )
insert ′〈:+:〉 iA uA iB uB (CTXSum (Inr cy)) = do y ← iB cy

return (Inr y)
insert ′〈:*:〉 iA uA iB uB (CTXProd (Inl (cx :*: y))) = do x ← iA cx

return (x :*: uB y)
insert ′〈:*:〉 iA uA iB uB (CTXProd (Inr (x :*: cy))) = do y ← iB cy

return (uA x :*: y)
insert ′〈Con c〉 iA uA (CTXCon cx ) = do x ← iA cx

return (Con x )
insert ′〈Label l〉 iA uA (CTXLab cx ) = do x ← iA cx

return (Label x )

Note that the extraction and insertion is happening in the application of the
generic abstraction to the Rec case (such as insert ′Rec and extract ′rec): the
helper functions only pass on the results.

Since up〈f 〉 · down〈f 〉 = id on locations in which it is possible to go down,
we expect similar equalities for the functions first , extract , and insert . We have
that the following computation

do {(t , c′)← first〈f 〉 ft c;
c′′ ← extract〈f 〉 c′;
ft ′ ← insert〈f 〉 c′ t ;
return (c c′′ ∧ ft ft ′ ) }

returns True on locations in which it is possible to go down.

Function right. The function right moves the focus to the next sibling to the
right in a tree, if it exists. The context is moved accordingly. The instance of
right on the data type Tree has been given in Section 1. The function right
satisfies the following property:

∀l . right〈f 〉 l 6= l =⇒ (left〈f 〉 · right〈f 〉) l = l ,

that is, first going right in the tree and then left again is the identity function on
locations in which it is possible to go to the right. Of course, the dual equality
holds on locations in which it is possible to go to the left.

Function right is defined by pattern matching on the context. It is impossible
to go to the right at the top of a value. Otherwise, we try to find the right sibling
of the current focus.

right〈f :: ?→ ?〉 :: Loc〈f 〉 → Loc〈f 〉
right〈f 〉 (t , c) = case out c of

Nothing → (t , c)
Just c′ → case next〈f 〉 t c′ of

Just (t ′, c′′)→ (t ′, in (Just c′′))
Nothing → (t , c).

The helper function next is a type-indexed function that returns the first location
that has the recursive value to the right of the selected value as its focus. Just as
there exists a function left such that left〈f 〉 · right〈f 〉 = id (on locations in which
it is possible to go to the right), there exists a function previous, such that

do {(t ′, c′)← next〈f 〉 t c;
(t ′′, c′′)← previous〈f 〉 t ′ c′;
return (c c′′ ∧ t t ′′)}

returns True (on locations in which it is possible to go to the right). We will give
the heading of function next , and omit the definitions of next ′ and previous.

next〈f :: ?→ ?〉 :: Fix f → Ctx 〈f 〉 (Fix f ) c → Maybe (Fix f ,Ctx 〈f 〉 (Fix f ) c)
next〈f 〉 t c = next ′〈f 〉 next ′Rec

extract ′Rec
insert ′Rec
first ′Rec
id id
c

next ′Rec t = Just t
dependency next ′ ← next ′ extract ′ insert ′ first ′ mkid unid

The dependency shows that next ′ is a rather complicated function that depends
on five other generic functions. This is reflected in its type:

type Next〈〈?〉〉 t a c = Ctx 〈t〉 → Maybe (a,Ctx 〈t〉)
type Next〈〈κ→ ν〉〉 t a c = ∀u .Next〈〈κ〉〉 u a c →

Extract〈〈κ〉〉 u c →
Insert〈〈κ〉〉 u →
First〈〈κ〉〉 u a c →
MkId〈〈κ〉〉 u →
UnId〈〈κ〉〉 u → Next〈〈ν〉〉 (t u) a c

The definition of function next ′ can be found in [24].

Exercise 5. (Easy exercise about constructor selection.) If we don’t want to use
the zipper, we can also keep track of the path to the current focus. Suppose we
want to use the path to determine the name of the top constructor of the current
focus in a value of a data type. The path determines which child of a value is
selected. Since the products used in our representations of data types are binary,
a path has the following structure:

data Dir = L | R
type Path = [Dir ]

Function selectCon〈〉 takes a value of a data type and a path, and returns the
constructor name at the position denoted by the path. For example,

data List = Nil | Cons Char List
?selectCon〈List〉 (Cons 2 (Cons 3 (Cons 6 Nil))) [R,R,R ]
"Nil"

data Tree = Leaf Int | Node Tree Int Tree
? selectCon〈Tree〉 (Node (Leaf 1 ) 3 (Node (Leaf 2) 1 (Leaf 5))) [R,R ]
"Node"

? selectCon〈Tree〉 (Node (Leaf 1 ) 3 (Node (Leaf 2) 1 (Leaf 5))) [R,R,L ]
"Leaf"

Define the type-indexed function selectCon〈〉, together with its kind-indexed
type.

Exercise 6. (Difficult exercise, in which you need (dual versions of) most of the
functions defined in this section.) Define the function left , which takes a location,
and returns the location to the left of the argument location, if possible.

Exercise 7. (Rather complicated exercise about an alternative representation of
values for editing purposes.) For several applications we have to extend a data
type such that it is possible to represent a place holder. For example, from the
data type Tree defined by

data Tree a = Leaf a | Node (Tree a) (Tree a)

we would like to obtain a type isomorphic to the following type:

data HoleTree a = Hole | Leaf a | Node (HoleTree a) (HoleTree a)

– Define a type-indexed data type Hole that takes a data type and returns a
data type in which also holes can be specified. Also give the kind-indexed
kind of this type-indexed data type. (The kind-indexed kind cannot and does
not have to be defined in Generic Haskell though.)

– Define a type-indexed function toHole which translates a value of a data
type t to a value of the data type Hole〈t〉, and a function fromHole that
does the inverse for values that do not contain holes anymore, so:

toHole〈t :: κ〉 :: ToHole〈〈κ〉〉 t
fromHole〈t :: κ〉 :: FromHole〈〈κ〉〉 t

type ToHole〈〈?〉〉 t = t → Hole〈t〉
type FromHole〈〈?〉〉 t = Hole〈t〉 → t

5 Conclusions

We have developed three advanced applications in Generic Haskell. In these
examples we use, besides type-indexed functions with kind-indexed kinds, type-
indexed data types, dependencies between and generic abstractions of generic
functions, and default and constructor cases. Some of the latest developments of
Generic Haskell have been guided by requirements from these applications.

We hope to develop more applications using Generic Haskell in the future,
both to develop the theory and the language. Current candidate applications are
more XML tools and editors.

sor. Dave Clarke, Andres Löh, Ralf Lämmel, Doaitse Swierstra and Jan de Wit
commented on or contributed to (parts of) previous versions of this paper.

References

1. Serge Abiteboul, Peter Buneman, and Dan Suciu. Data on the Web. Morgan
Kaufmann Publishers, 2000.

2. Altova. XML Spy. Whitepaper available from http://www.xmlspy.com, 2002.
3. The Apache Project. Cocoon. Available from http://xml.apache.org/cocoon/,

2002.
4. R. Backhouse, P. Jansson, J. Jeuring, and L. Meertens. Generic programming: An

introduction. In S. Doaitse Swierstra, Pedro R. Henriques, and José N. Oliveira,

9. Sophie Cluet and Jérôme Siméon. YATL: a functional and declarative language
for XML, 2000.

10. Richard H. Connelly and F. Lockwood Morris. A generalization of the trie data
structure. Mathematical Structures in Computer Science, 5(3):381–418, September
1995.



42 R. Hinze, J. Jeuring

11. X-Hive Corporation. X-Hive. Available from http://www.xhive.com, 2002.
12. XMLSolutions Corporation. XMLZip. Available from http://www.xmlzip.com/,

1999.

